	Filename statement defines a link between a given symbolic reference name (POP)  called fileref  and the physical path and name of the external (non SAS) file.
	filename pop '/folders/myshortcuts/My_Folders/flat/samplefile-pop.csv';

	%let assigns a value to macro variables, to be used later in the code.
It enables easy adaptation to changes need over time and preferences.

	%let from_year = 2000;
%let upto_year = 2007;
%let prefix = pop; 

	Data defined the output dataset(s) to be created
	data want;

	Infile defines the input external file to read.  In this case I use the predefined fileref  with some options: truncover  and firstobs.  (*1)
	infile pop truncover firstobs=2;

	Input reads one record from the input file. The a_line is just a variable name I choosd.  $ tells sas to read a string of characters, within a given upper limit length (120). $120. Is called informat and the dot at the end is desired and part of it.
	input a_line $120.;

	Scan function scans the string, defined by first argument (i.e. a_line, the record just read) and enters the nTH. Substring defined by 2nd argument that is between delimiter(s) given in third argument. 
So city is the 1st string and state is the 2nd string between commas, assuming none of them contains a comma inside.
	    city = scan(a_line,1,',');
    state = scan(a_line,2,',');

	Index is a function that returns the position of a substring given as 2nd argument in the string given as 1st argument. If substring not found it returns zero. In this case the 2nd argument is given with function strip in order to ignore trailing blanks of the variable state.
	pos = index(a_line,strip(state));

	1st population shall start 2 characters after the state – skipping the comma and the opening double quotes.
	pos = pos + lengthn(state) +2; 


	Population variables, in your csv file, are given in double quotes and contains commas (like: "1,234,567","2,678,987"). Compress function takes of the characters given by the 2nd argument from the string given by 1st argument.
Substr function results in a sub string starting position given by 2nd argument to the end (or to a given 3rd argument).
	popx = compress(substr(a_line,pos),',');

	Array defines array by given list of variables. In this case I created the variable names as concatenation of prefix and a number given by the macro variables at the top of this program. Pay attention how I use macro variables, e.g. &prefix. Strats with & then the macro variable name then a dot to assign the end of it, in order to enable concatenation of more charactrs.
	array pop &prefix.&from_year - &prefix.&upto_year;

	Doing a loop 8 times as you have 8 years of population.
Each time the population number is given between double quote as delimiters.

For testing only, I used put function to write selected variables to the LOG.

Finaly, after removing the commas and the quotes I use the function input to convert the population into numeric variables.
	do i=1 to 8;
   vx = scan(popx,i,'"'); 
   put i= vx=;
   pop(i) = input(vx,comma12.);
end;


	Keep defines which variables should be written to output.
Otherwise I could use drop to define which variables to drop, that is not write to the output
	keep city state &prefix.&from_year - &prefix.&upto_year;

	RUN; statement tells SAS that the code is ready to be executed.
	RUN;

	
	


[bookmark: _GoBack]
